


8051-es Mikrokontroller- és Assembler-tanfolyam

6. rész: A soros interfész programozása



Már a Compuboard első használatbavétele óta használjuk az EMON51-es monitorral a 8051 által a soros kommunikáció céljára rendelkezésre bocsátott lehetőségeket. Most eljött az ideje, hogy az interfész kulisszái mögé pillantsunk és azt magunk programozzuk.

A 8051 soros interfésze

A soros interfész a 8051-es mikrovezérlőknél a legkompaktabb „On-Chip-Peripheral”. Az alapkursusnak ez a fejezete így a tanfolyam legnehezebb részei közé tartozik, ezért célszerű igen gondosan feldolgozni. Aki később, saját építési projektjei során 8051-esének a külvilággal való kapcsolatát V24-es interfész útján kívánja megvalósítani, annak igencsak neki kell veselkednie még akkor is, ha a példák kapcsán az EMON51 monitor egy sor segítséget nyújt.

A soros interfész számos olyan üzemmódra alkalmas, melyek elsősorban a 8051-es hálózatok kiépítésére szolgálnak és bennünket a tanfolyam során nem érdekelnek. Mi a legegyszerűbb üzemmódra koncentrálnunk, ez pedig a start- és stopbittel ellátott (paritásbit nélküli) 8-bites adatok aszinkron adása és vétele. Ebben az üzemmódban egyrészt tudunk a PC-vel kommunikálni, másrészt MIDI adatforgalom is megvalósítható.

A soros interfész vezérlésére és üzemmódjának meghatározására a 09H₁₆ című SCON (Serial Control) speciális funkcióregiszter szolgál. Az egyes bitek jelentéseit az 1. ábrán foglaltuk össze. Az általunk kívánt üzemmód beállítása céljából SM0=0 és SM1=1 értékeket kell beírni. Ebben az 1-es módnak nevezett üzemmódban az SM2 bit az átviteli hibák felismerésének igénye szerint kapcsol-

ható (lásd később). A további bitek jelentéséről ugyancsak később fogunk még beszélni.

Vevő- és adópuffer

Soros adatok adására és vételére a 8051-en belül egy léptetőregisztert használunk. Adás során ezt a regisztert párhuzamosan töltjük fel adatokkal. Ezután a biteket a (Baudsebességnek nevezett) adatátviteli sebességgel toljuk ki a léptetőregiszterből. Vétel esetén az adatbitek sorosan kerülnek a léptetőregiszterbe és az összes bit vétele után párhuzamosan kerülnek kiolvasásra. Vételi- és adópufferként az SBUF jelölésű (09H₁₆ című) SFR szolgál. Ez mintegy megkettőzve áll rendelkezésre, nevezetesen egyszer vevő- és egyszer olvasópufferként. Attól függően, hogy az SBUF regiszter kiolvasásra

ség generálására szolgáló kapcsolási részek. A Baudsebesség (bitléptető órajel) generálására az 1-es Timer vagy (csak a 8052-nél) a 2-es Timer túlszorzása vehető igénybe. Az SMOD jelzésű kapcsolóval egy 1:2 arányú előosztó iktatható be. A vevő- és az adó órajel kiválasztására az RCLK és (csak a 8052-nél, illetve 8032-nél) a TCLK kapcsoló szolgál. Az így előkészített órajelek a TX-CONTROL adóvezérlésben, illetve az RX-CONTROL vevővezérlésben további, 16-szoros osztásra kerülnek.

Az adó léptetőregisztert, melynek kimenete TXD jelöléssel a 8051-es 11-es kivezetésén található, az adásvezérlő látja el órajellel és tölti fel adatokkal (az ábra felső része). Az adatok a 8051-es belső adatbuszáról párhuzamosan kerülnek az SBUF regiszterbe. A ZERO-DETECTOR állapítja meg, hogy mikor fejeződött be az összes bit leadása és azt jelenti az adásvezérlőnek, amely ezután egy TI soros Port-Interruptot tud kioldani. Ennek során megtörténik a TI bit (SCON.1) beírása.

tőregiszterbe kerülnek. Utolsóként a stopbit kerül beolvasásra (a vevő léptetőregiszter szélessége 9 bit). Ezután a vett nyolc bittel feltöltődik az SBUF regiszter és sor kerül a „vevő megtelt” bit (SCON.0 =RI) beírására. Ez a folyamat azonban a következő két esetben nem játszódik le:

- akkor ha RI beírása előzőleg már megtörtént, azaz a vett adatok átvételére nem került sor, és
- ha SM2=1 és a stopbitnek nem volt 1 értéke. SM2=1 értékével tehát azt lehet kiváltani, hogy Framing-Error (hibás stopbit) esetén a bájtok ne kerüljenek átvételre.

Az RI bit segítségével megállapítható, hogy az SBUF-ben készen áll-e elvitelre egy vett bájti. A vett bájti azután az SBUF regiszterből olvasható ki.

A két interrupt esemény, TI (Transmitter-Interrupt) és RI (Receiver Interrupt) egyébként ugyanazt az interruptot váltja ki és az interrupt szoftvernek ezért önmagának kell megállapítania, hogy az interruptot egy korrekten leadott jel vagy egy

		(MSB)				(LSB)			
		SM0	SM1	SM2	REN	TB8	RB8	TI	RI
Where SM0, SM1 specify the serial port mode, as follows:									
SM0	SM1	Mode	Description	Baud Rate					
0	0	0	shift register	$f_{osc}/12$					
0	1	1	8-bit UART	variable					
1	0	2	9-bit UART	$f_{osc}/64$ or $f_{osc}/32$					
1	1	3	9-bit UART variable						
• SM2		enables the multiprocessor communication feature in Modes 2 and 3. In Mode 2 or 3, if SM2 is set to 1 then RI will not be activated if the received 9th data bit (RB8) is 0. In Mode 1, if SM2 = 1 then RI will not be activated if a valid stop bit was not received. In Mode 0, SM2 should be 0.							
• REN		enables serial reception. Set by software to enable reception. Clear by software to disable reception.							
		• TB8		is the 9th data bit that will be transmitted in Modes 2 and 3. Set or clear by software as desired.					
		• RB8		in Modes 2 and 3, is the 9th data bit that was received. In Mode 1, if SM2 = 0, RB8 is the stop bit that was received. In Mode 0, RB8 is not used.					
		• TI		is transmit interrupt flag. Set by hardware at the end of the 8th bit time in Mode 0, or at the beginning of the stop bit in the other modes, in any serial transmission. Must be cleared by software.					
		• RI		is receive interrupt flag. Set by hardware at the end of the 8th bit time in Mode 0, or halfway through the stop bit time in the other modes, in any serial reception (except see SM2). Must be cleared by software.					

1. ábra. A bitek elhelyezkedése a SCON adásvezérlő SFR-ben

vagy beírásra kerül-e, vételi vagy adópufferként kell kezelni. Ennek megfelelően adásra és vételre külön-külön egy léptetőregiszter áll rendelkezésre, azért, hogy az adás és a vétel tényleg egyidejűleg történhessék (teljes duplex üzemmód).

A funkciók áttekintése

A 8051-ben található soros interfész belső kapcsolásának áttekintése a 2. ábrán található. Az ábra bal felső sarkában felismerhetők a Baudsebes-

ség vezérlő az ábra alsó részén szerepel. A soros adatok a 8051-es 10-es kivezetésén át jutnak be a vevő léptetőregiszterbe. Az adatok egyidejűleg az 1-TO-O-TRANSITION DETECTOR-ba jutnak, mely a soros adatátvitelnél a startbit felismerésére és a vételi órajelnek megfelelően a beérkező bitekhez való szinkronizálására szolgál. A vételi folyamatot az RXD bemeneten megjelenő visszafutó éi (startbit a 8051-es 10-es kivezetésén) indítja. A bejövő bitek az órajelnek megfelelően a lépte-

vett jel váltotta-e ki. A soros interfész interruptjai csak akkor kerülnek kioldásra, ha az Interrupt-Enable-Registerben a megfelelő IE.4 bit (vesd össze a tanfolyam legutóbbi résszel) be van billentve.

A TXD és RXD vonalak kapcsolása a Compuboardon és a bővítményen a 3. ábra szerinti. Ott azt is bemutatjuk, hogy a különböző jumperok útján hogyan köthető össze a PC-vel való kommunikációra szolgáló 9 pólusú csatlakozó vagy a MIDI interfész a 8051 soros interfészével.

az 1-es Timert Auto-Preload módban belső órajellel (itt 1 MHz) használjuk, ez a 22-es sorban található utasítással történik. Preload értéként 243-mat használunk (4-es, 24-es és 25-ös sor). Ezáltal az 1-es Timer osztási arányára $256-243=13$ adódik. SMOD-ra 1-et írunk be és ezzel a 1:2 arányú osztó kikapcsolódik (21-es sor). A kialakuló Baudsebesség: $1 \text{ MHz} / 13 / 16 = 4807,6923 \dots$ bit/szekundum. Ez nem pontosan 4800 Baud, de nekünk elég pontos. Ahhoz, hogy az 1-es Timer szabadonfutóvá válhassék, be kell még billenteni a TCON Timer-Kontrollregiszterben (vö. a legutóbbi folytatásunkkal) a 6-os bitet (26-os sor). A bit-órajel ezzel rendelkezésre áll.

Az előbbiekkal kapcsolatban egy megjegyzés: 8052 vagy 8032 esetén Baudgenerátorként a 2-es Timert is használhatjuk. Ez a 2-es Timer T2CON vezérlőregiszterében a TCLK, illetve RCLK bitek beírása útján történik.

Mivel azonban T2CON a resetelés után 00_{16} értéken áll, induláskor Baudgenerátorként a 8052-es mindig az 1-es Timerre kapcsol és így a 8051-re írt programok a 8052-n is hibátlanul futtathatók.

12 MHz-es kristállyal pontosan 1200 vagy 4800 Baud nem generálható. Ez az oka annak, hogy miért szeretik annyira a 8051-est 11,0592 MHz-es kristállyal üzemeltetni. Ezáltal ugyanis lehetővé válik a szokásos órajel sebességek egésszámú osztás útján történő generálása. Ebben az esetben viszont egy utasítás végrehajtása már nem pontosan 1 mikroszekundumot vesz igénybe, mint a 12 MHz-es órajelnél, hanem $0,9044225$ mikroszekundumot. A többi időzítés tehát ennek megfelelően bonyolultabbá válik.

Ez a dilemma vezetett egyébként ahhoz, hogy a 8051-es néhány későbbi modelljébe kiegészítő Baudgenerátorokat integráltak. A 80535 és a 80537 például egy olyan

további Baudgenerátort tartalmaz, mely alkalmas 12 MHz-ből pontosan 4800 bit/szekundum és 9600 bit/szekundum generálására.

Adás

Egy bájt adásának indítása igen egyszerű módon történik. A bájtot egyszerűen be kell írni az SBUF regiszterbe. A beírási utasítás hatására a soros adásvezérlő indítja a soros adási folyamatot. Az adóregiszter egy kilencedik bittel is rendelkezik, mely az adási folyamat kezdetén automatikusan „1”-gyel kerül feltöltésre. Ezt követően először a startbit („0”=Low) adására kerül sor (vö. a 3. ábrával).

Ezután következik a nyolc adatbit egymást követően történő leadása, a 0-ás bittel kezdődően. Ennek során az adó SBUF regisztere nullákkal kerül feltöltésre. Amikor a kilencedik bit – tehát az „1” értékű stopbit – adása megtörtént akkor az átvitel befejezésének jelzése céljából az

SCON TI bitjére „1”-es kerül beírásra. Ezáltal, a megfelelő bitnek az Interrupt-Enable-Regiszterbe történő beírása útján, interrupt váltható ki. Ez a bit azonban az átvitel végének megállapítása céljából szoftver úton is lekérdezhető (Polling). Az EMON51-ben egy (az akkumulátorban található) karakter adásáért az SND alprogram (30...39-es sor) a felelős. Ez a rutin mindaddig várakozik, míg a SCON regiszter 1-es bitjének értéke „0”. Mivel ebben a bitben a 0 érték azt jelzi, hogy pillanatnyilag még az előző karakter továbbítása folyik, ez a várakozás feltétlenül szükséges is. Ha ez a (TI) bit „1” értéket vesz fel, az azt jelenti, hogy a karakter adása befejeződött és sor kerülhet a következő bit továbbítására. Ez úgy történik, hogy a 31-es sorban bekövetkezik a TI bit újbóli nullázása és a 32-es sorban a továbbításra szánt karakter az SBUF adópufferbe kerül átírásra. Ettől kezdve egészen a karakter adásának befejezéséig a TI bit 0 értékű marad. A processzor azonban

4. ábra. V24-es programozás az EMON51-ben

```

4
***** LISTING of EASM51 (V24BSP) *****
LINE LOC OBJ          T          SOURCE
1 0000                : entnommen aus EMON51.A51
2 0000                :
3 0000                :
4 0000                : V24SPD EQU 256-13 ; V24 Geschwindigkeit: 1MHz /16/13 = 4807.69 Baud
5 0000                :
6 0000                : ; SFRs :
7 0000                : PSW EQU 0D0H
8 0000                : ACC EQU 0E0H
9 0000                :
10 0000               : PCON EQU 087H
11 0000               : TCON EQU 088H
12 0000               : TMOD EQU 089H
13 0000               : TL1 EQU 08BH
14 0000               : TH1 EQU 08DH
15 0000               : SCON EQU 098H
16 0000               : SBUF EQU 099H
17 0000               :
18 0000               : CNT1 EQU 050H ; Im RAM : Zaehler fuer CRLF Zeit
19 0000               :
20 0000               : ORG 0
21 0000 75 87 80 [2] V24SET MOV PCON,#080H ; SMOD=1
22 0003 75 89 22 [2]      MOV TMOD,#22H ; beide Zaehler als TIMER
23 0006                : ; TCLK=0 ,RCLK=0 bei 8051 und auch in T2CON
24 0006 75 8D F3 [2]      MOV TH1,#V24SPD ; Preload Wert TIMER1 (Baudraten-generator)
25 0009 75 8B F3 [2]      MOV TL1,#V24SPD
26 000C D2 8E [1]        SETB TCON.6 ; STARTE Zaehler 1
27 000E 75 98 52 [2]      MOV SCON,#052H ; MODE 1 , REN=1 , TI=1 , RI=0
28 0011                : usw.
29 0011                :
30 0011 30 99 FD [2] SMD JNB SCON.1,SMD ; warte bis letztes Zeichen endlich weg
31 0014 C2 99 [1]        CLR SCON.1 ; TI = 0
32 0016 F5 99 [1]        MOV SBUF,A ; starte Sendevorgang
33 0018 B4 0A 0D [2]     CJNE A,#10,OK2 ; ist ein LF gesendet worden ?
34 001B 75 50 64 [2]     WAITCR MOV CNT1,#100 ; wenn ja warte wg. langsamer Terminals
35 001E 74 FF [1]        LOP1 MOV A,#255
36 0020 D5 E0 FD [2]     LOP2 DJNZ ACC,LOP2
37 0023 D5 50 F8 [2]     DJNZ CNT1,LOP1
38 0026 74 0A [1]        MOV A,#10 ; restauriere A
39 0028 22 [2]          OK2 RET ; alles ok
40 0029                :
41 0029                : GETCHR EQU $ ; hole Zeichn von V24 Schnittstelle
42 0029 30 98 FD [2]     GETC1 JNB SCON.0,GETC1 ; warte bis eins da ist
43 002C C2 98 [1]        CLR SCON.0 ; signalisiere : abgeholt
44 002E E5 99 [1]        MOV A,SBUF ; hole es aus Puffer
45 0030 22 [2]          [1] RET ; fertig
46 0031                :
47 0031 20 98 03 [2]     TSTC JB SCON.0,istda ; Ist ein Zeichen da ?
48 0034 74 00 [1]        MOV A,#0 ; nein
49 0036 22 [2]          RET
50 0037 74 01 [1]        istda MOV A,#1 ; ja
51 0039 22 [2]          RET
52 003A                :
53 003A                : END
***** SYMBOLTABLE (21 symbols) *****
V24SPD :00F3 PSW :00D0 ACC :00E0 PCON :0087
TCON :0088 TMOD :0089 TL1 :008B TH1 :008D
SCON :0098 SBUF :0099 CNT1 :0050 V24SET :0000
SMD :0011 WAITCR :001B LOP1 :001E LOP2 :0020
OK2 :0028 GETCHR :0029 GETC1 :0029 TSTC :0031
istda :0037
910109-7-14

```

4. ábra. entnommen aus EMON51.A51: az EMON51.A51-ből véve V24 Geschwindigkeit: V24-es sebessége Im RAM. Zaehler fuer CRLF ZEIT: a RAM-ban a CRLF idő számlálója beide Zaehler als TIMER: mindkét számláló TIMER-ként TCLK=0, RCLK=0 bei 8051 und auch in T2CON: TCLK=0, RCLK=0 8051-nél és T2CON-ban is Preload Wert TIMER1 (Baudraten generator): TIMER1 (Baudsebesség generator) Preload értéke STARTE Zaehler 1: INDÍTSD az 1-es számlálót warte bis letztes Zeichen endlich weg: várj, míg az utolsó karakter kimegy starte Sendevorgang: indítsd az adási folyamatot ist ein LF gesendet worden?: LF adása történt? wenn ja warte wg. langsamer Terminals: ha igem, várj a lassúbb terminálok miatt restauriere A: állítsd vissza A-t alles ok: minden ok hole Zeichen von V24 Schnittstelle: hozd el a karaktert a V24 interfészről signalisiere: abgeholt: jelezd, hogy elhozva hole es aus Puffer: hozd ki a pufferből fertig: kész Ist ein Zeichen da?: Van itt karakter? nein: nincs ja: igen

eközben már tovább működik. Az EMON51-ben most egy további komplikációt is számításba kell venni: a terminálok (vagy a terminálprogramok) egy részének a soremelés (LF=LINEFEED = 10₁₀=0A₁₆) után a képernyő tartalmának csúsztatásához (scrolling) bizonyos időre van szüksége. Ennek letelte előtt további karaktert nem szabad leadni.

5. ábra

MIDI Kanal 1: MIDI 1. csatorna
Bide Auto Preload Timer: Mindkét Auto Preload Timer
Preload Wert fuer TIMERO: Preload érték TIMERO-hoz
TIMEO starten: TIMERO-t indítani
Tasten einlesen: Nyomógomb beolvasás

starte wenn Taste gedruickt: indíts, ha a gomb meg van nyomva
Zeiger auf NOTEN: A mutatót a NOTEN (hangjegyek)-re állítani
sende an MIDI (MIDI KANAL): adj a MIDI felé (MIDI CSA-TORNA)
(MIDI STAERKE): (MIDI HANGERŐ)
Zeitdauer: időtartam
=0 heisst Ende: =0 azt jelenti, hogy vége
sonst diese Zeit warten: különben ezt az időt kívárni
und weiter bei naechster Note: és a következő hangjegynél folytatni
Byte holen: bájt elhozni
Datenzeiger erhoehen: adatmutatót növelni
fertig: kész
ca ACCx100x200x2 Mikrosekunden warten: kb. ACC x100x200x2 mikroszekundumot várni

Warten bis Sender leer: várni, míg az adó kiürül
Bit loeschen: bitet törölni
Byte holen: bájt kihozni
Ausgeben: kiadni
Warten bis Empfaenger voll: várni, míg a vevő megtelik
Bit loeschen: bitet törölni
Byte holen: bájt kihozni
Note 48 mit 100 an, 6 Zeiten warten: 48-as hangjegyet 100-as hangerővel bekapcsolni, 6 egységnyi időt várni
Note 48 wieder aus 1 Zeit warten: 48-as hangjegyet újra kikapcsolni, 1 egységnyi időt várni
Note 52 an mit 100: 52-es hangjegyet 100-as hangerővel bekapcsolni
usw.: stb.
Laenge 0 heisst Ende: 0 hossz azt jelenti, vége

6. ábra
Initialisiere TIMER 1 als Baudratengenerator 31,25 kbit/sec: Inicializáld az 1-es TIMER-t 31,25 kbit/sec-os Baudsebesség generátorként
Initialisiere Schnittstelle: Inicializáld az interfészt
Taste gedruickt?: Gomb megnyomva?
ja: igen
nein: nem
Initialisiere DPTR: Inicializáld a DPTR-t
SENDE: Midikanal, Note, Stärke aus Tabelle: Add: MIDI-csatornát, hangjegyet, hangerősséget a táblázatból
hole Dauer aus Tabelle: hozd az időtartamot a táblázatból
Dauer=0: Időtartam=0
ja: igen
nein: nem
Ende der Tabelle erreicht: Elértünk a táblázat végére
Warte Dauer: várd ki az időtartamot


5. ábra. A MIDI-Sequencer listája

```

5 ***** LISTING of EASMS1 (BSP13) *****
LINE LOC OBJ T SOURCE
1 0000 ; ***** DATEI BSP13.A51 *****
2 0000 ;
3 0000 ACC EQU 0E0H
4 0000 PCOM EQU 087H
5 0000 TCOM EQU 088H
6 0000 TMOD EQU 089H
7 0000 TL1 EQU 08BH
8 0000 TH1 EQU 08DH
9 0000 SCOM EQU 098H
10 0000 SBUF EQU 099H
11 0000 ;
12 0000 CH EQU 090H ; MIDI Kanal 1
13 0000 ;
14 0000 ORG 4100H
15 4100 75 87 80 [2] MOV PCOM,#10000000B ; SMOD=1
16 4103 75 89 22 [2] MOV TMOD,#00100010B ; Beide Auto Preload Timer
17 4106 75 8D FE [2] MOV TH1,#256-2 ; Preload Wert fuer TIMERO
18 4109 75 8B FE [2] MOV TL1,#256-2
19 410C D2 8E [1] SETB TCOM.6 ; TIMERO starten
20 410E 75 98 52 [2] MOV SCOM,#01010010B ; MODE 1 , Enable receiver
21 4111 90 C0 00 [2] START1 MOV DPTR,#0C000H
22 4114 E0 [2] MOVX A,@DPTR ; Tasten einlesen
23 4115 20 E7 F9 [2] JB ACC.7,START1 ; starte wenn Taste gedruickt
24 4118 90 41 4F [2] MOV DPTR,#NOTES ; Zeiger auf NOTEN
25 411B 74 90 [1] NOTELP MOV A,#CH
26 411D 31 3F [2] ACALL SMDMIDI ; sende an MIDI ( MIDI KANAL )
27 411F 31 2F [2] ACALL GETWXT
28 4121 31 3F [2] ACALL SMDMIDI ; ( MIDI NOTE )
29 4123 31 2F [2] ACALL GETWXT
30 4125 31 3F [2] ACALL SMDMIDI ; ( MIDI STAERKE )
31 4127 31 2F [2] ACALL GETWXT ; Zeitdauer
32 4129 60 E6 [2] JZ START1 ; =0 heisst Ende
33 412B 31 33 [2] ACALL WAIT ; sonst diese Zeit warten
34 412D 80 EC [2] SJMP NOTELP ; und weiter bei naechster Note
35 412F ;
36 412F E4 [1] GETWXT CLR A ; Offset 0
37 4130 93 [2] MOVC A,#A+DPTR ; Byte holen
38 4131 A3 [2] INC DPTR ; Datenzeiger erhoehen
39 4132 22 [2] RET ; fertig
40 4133 ;
41 4133 7F 64 [1] WAIT MOV R7,#100 ; ca ACCx100x200x2 Mikrosekunden Warten
42 4135 7E C8 [1] WAIT1 MOV R6,#200
43 4137 DE FE [2] WAIT2 DJNZ R6,WAIT2
44 4139 DF FA [2] DJNZ R7,WAIT1
45 413B D5 E0 F5 [2] DJNZ ACC,WAIT
46 413E 22 [2] RET
47 413F ;
48 413F 30 99 FD [2] SMDMIDI JNB SCOM.1,SMDMIDI ; Warten bis Sender leer
49 4142 C2 99 [1] CLR SCOM.1 ; Bit loeschen
50 4144 F5 99 [1] MOV SBUF,A ; Ausgeben
51 4146 22 [2] RET
52 4147 ;
53 4147 30 98 FD [2] GETMIDI JNB SCOM.0,GETMIDI ; Warten bis Empfaenger voll
54 414A C2 98 [1] CLR SCOM.0 ; Bit loeschen
55 414C E5 99 [1] MOV A,SBUF ; Byte holen
56 414E 22 [2] RET
57 414F ;
58 414F 30 64 06 NOTES DB 48,100,6 ; Note 48 mit 100 an , 6 Zeiten warten
59 4152 30 00 01 DB 48,0 ,1 ; Note 48 wieder aus , 1 Zeit warten
60 4155 34 64 06 DB 52,100,6 ; Note 52 an mit 100
61 4158 34 00 01 DB 52,0 ,1 ; wieder aus
62 415B 37 64 0A DB 55,100,10 ; usw.
63 415E 37 00 00 DB 55,0 ,0 ; Laenge 0 heisst Ende
64 4161 END
***** SYMBOLTABLE (18 symbols) *****
ACC :00E0 PCOM :0087 TCOM :0088 TMOD :0089
TL1 :008B TH1 :008D SCOM :0098 SBUF :0099
CH :0090 START1 :4111 NOTELP :411B GETWXT :412F
WAIT :4133 WAIT1 :4135 WAIT2 :4137 SMDMIDI :413F
GETMIDI :4147 NOTES :414F

```

6. ábra. A MIDI-Sequencer folyamatábrája



Ezért az EMON51 a 33-as sorban megvizsgálja, hogy nem egy LINEFEED adása történt-e utoljára. Ha igen, akkor a WAITCR várakozóhurokban mintegy 100 · 255 · 3 µs = 76,5 ms várakozásra kerül sor. Ez az idő a terminálok legnagyobb része számára elegendő.

Ezzel az EMON51 monitor SND adórutinjának leírása befejeződött. Ahhoz, hogy a rutin helyesen működjék, a TI bitet a program kezdetén „1”-re kell beállítani (házi feladat az Olvasónak: vajon miért?). Reszetteléskor azonban ez a bit nulla értéket vesz fel (vö. a tanfolyam 2. részének 4. ábrájával). Ha ezt az adás lehetőségét jelző bitként kívánjuk használni, akkor a program kezdetén „1”-et írunk bele.

Az SCON interfész-vezérlőszó az EMON51-ben

Most magyarázzuk meg a soros interfész üzemmódjának beállítására szolgáló 27-es sort a 4. ábrán. Az 52₁₆ = 01010010B vezérlőszó egyrészt az 1-es üzemmódból adódik. Ez meghatározza az 5-ös, 6-os és a 7-es bitet (=010B). Mivel adatokat akarunk venni, a Receiver-Enable bit (4-es bit) helyére „1”-et kell beírni. A 2-es és a 3-as bit számunkra egyelőre nem érdekes. Az 1-es bitet (TI-bit) azért billentjük be, hogy a további programozáshoz az SBUF regisztert üresnek jelöljük és ezzel egy további adási utasítás végrehajthatóvá válik. Mivel eddig bájt vételére nem került sor, az RI bitet (0-ás bit) „0”-ra állítjuk be. Ezzel a soros interfész 4800 Baudos duplex üzemmódra (8 bit egy start- és egy stopbit) történő inicializálását befejeztük.

Vétel

Az EMON51 programon belül a soros adatok vételét a GETCHR (Get Character) alprogram végzi el. Először (42-es sor) az alprogram mindaddig vár, míg az SCON regiszter 0-s bitjének (RI-bit) a bebillentése meg nem történik. Egy karakter vételének befejeződését ugyanis éppen ennek az RI-bitnek az „1”-es értéke jelzi. Végül a 43-as sorban a program az RI bitet ismét nullázza és a vett karaktert az SBUF vevőpufferből az akkumulátorba viszi át. És ezzel már mindent el is mondtunk.

Egyes esetekben szükséges lehet az az információ, hogy egy karakter átvételre rendelkezésre áll-e. Erre szol-

gál a TSTC alprogram (47-től 51-ig terjedő sorok), mely az akkumulátorba egy 1-et ír, ha egy karakter készen áll az átadásra. Egyébként az akkumulátorba a 0 érték kerül. Az alprogram működése az előbbi magyarázatok alapján világos: egyszerűen az RI bit értékelését végzi.

MIDI jelsorozatok adása (MIDI-Sequencer)

Most a soros interfészt adatoknak egy MIDI készülékhez való továbbítására kívánjuk használni. Programunk feladata egy billentyű (a többfunkciós panel S2 nyomógombja) megnyomásának hatására egy kottautasítás-sorozat leadása. Ezáltal például egy egyszerű dallam játszható le vagy egy Drumbox valósítható meg. Különböző dallamok könnyű programozhatósága érdekében a leadandó adatokat táblázatba célszerű összefoglalni. A program így egy kis zene-szerkesztő (Sequencer) programozásának kiindulópontját képezheti.

A MIDI kottautasítás három bájtól áll. Az első bájt annak közlésére szolgál, hogy kottautasítás következik és hogy a dallamot melyik csatornán kell lejátszani. Az általunk használt 1-es csatornára vonatkozóan (ennek első sorszáma 0), ennek a bájtának a 090₁₆ értéket kell felvennie. A következő bájt azt határozza meg, hogy az utasítás melyik hangjegyre vonatkozik, a harmadik bájt pedig a hangjegy lejátszásának erősségét adja meg (minden egyes hangot be- és ki- kell kapcsolni!). Minden utasítás továbbítása tehát olyan három bájt leadását kívánja meg, melyek közül az első bájt mindig azonos. Ezt az első bájtot ezért a táblázatban nem tároljuk. Ezzel szemben tárolunk a táblázatban harmadik bájtjéknél egy olyan időinformációt, amely azt adja meg, hogy a következő MIDI utasítás kiadásáig mennyi időnek kell elteltie. A táblázatban tehát minden egyes utasításhoz mindig a következő három bájt áll:

Hangjegy, Hangerő, Időtartam

Az időtartam=0 a táblázat végét jelöli. A listában (5. ábra) a táblázat a NOTES címkével kezdődően található meg. A táblázat egyes bájtjainak kiolvasása a (táblázatot tartalmazó) programtárolónak a DPTR mutatóval történő címzése útján történik. A főprog-

ram megértése egyszerű. A program folyamatábráját a 6. ábrán mutatjuk be.

Most térjünk rá a példa kapcsán a soros interfész programozására. A MIDI adatátvitel 31,25 kBit/szekundum sebességgel történik. A Baudsebesség tehát a (8051-es belső) órajel (1 MHz) 32-vel való osztásával generálható. Mivel a 8051-es adómechanizmusa önmaga végez egy 16-tal való osztást, a MIDI adatok továbbítása céljából az 1-es Timernek 2-vel való osztást kell végrehajtania és az SMOD-ba „1”-et kell beírni. Programunk ezt a 15...19 sorokban végzi el. Végül a 20-as sorban kerül sor a soros üzemmód paramétereinek beállítására (1-es módus, vevő bekapcsolva, TI=„1”, az adó tehát üres). Az üzemmód rögzítése és a so-

ros interfész inicializálása tehát – a Baudsebesség meghatározását kivéve – az EMON51 esetében követett eljárásnak felel meg.

A tulajdonképpeni adórutint az SNDMIDI alprogram képezi. Ez adja le azt a karaktert, amelyik az akkumulátorban áll. Először a TI-bit (a SCON-ban az 1-es bit) alapján megvizsgálja, hogy az adó a legutoljára adott bájtot teljes egészében leadta-e. Ha nem, akkor addig várakozik, míg ez meg nem történik. Ezután bekövetkezik az SBUF „Adóregiszter-Üres” bit (1-es bit) törlése és egy az SBUF Adópuffer-Regiszterbe történő beírás utasítással (50-es sor) indul a következő bájt adása. Maga az adórutin tehát ugyanaz, mint az EMON51 monitorban. ■

MIGÉRT RT

VÁLLALKOZÓK! VÁLLALATOK! OKTATÁSI INTÉZMÉNYEK!

Rendkívüli termékajánlat az Önök számára!
Új, korszerű, európai színvonalat képviselő műszerek!
Előnyös árfekvés, azonnali szállítás!

HUNG CHANG gyártmányú (Dél-Korea)

- ▷ OS-615 típusú 15 MHz, 2 csatornás telepes oszcilloszkóp
- ▷ HC-8204/a típusú hanggenerátor, 20 Hz–200 kHz
- ▷ MODE 300/A típusú analóg lakatfogó 6–300 A
- ▷ HC-5050 E típusú analóg kéziműszer
- ▷ HC-26 típusú digitális kéziműszer
- ▷ DM-301 digitális kéziműszer
- ▷ PB-1 Power Bank (tölthető, hordozható univerzális áramforrás)
- ▷ HC-2020 S analóg műszer
- ▷ G 305 funkciógenerátor
- ▷ 8100 digitális frekvenciamérő
- ▷ 5502 típusú 20 MHz kétcsatornás oszcilloszkóp

Megvásárolható:

1. sz. Műszerszaküzlet
1061 Bp., Andrásy út 2.
Telefon: 132-2936



ATLANTIC